C in the middle, t_i becomes t. And a = $ue^{-\epsilon c}$. And finally $\eta = -\epsilon \frac{1+t}{1-t}$

```
ϵ /: ϵ^n_ /; n > 1 := 0;
PBWBasis = {a, c, w};

B[U@c, U@u] = - (B[U@u, U@c] = -U@u);
B[U@u, U@w] = - (B[U@w, U@u] = (1 - t) U[] + ϵ U[u, w] - 2 ϵ t U[c]);

B[U@c, U@w] = - (B[U@w, U@c] = U@w);
B[U@c, U@a] = - (B[U@a, U@c] = -U@a);
B[U@a, U@w] = - (B[U@w, U@a] = (1 - t) U[] - ϵ (1 + t) U[c]);
(*
B[U@a,U@w]=-(B[U@w,U@a]= (1-t)U[]+η(1-t) U[c]);
*)
```

```
(*Setting a = ue^-ϵc gives the bracket as above*)
U@w ** U@u ** (U[] - ϵ U[c]) -
  U@u ** (U[] - ϵ U[c]) ** U[w] - ((1 - t) U[] - ϵ (1 + t) U[c]) // Simp
```

```
0
```

```
UU[l___, x_^n_, r___] := UU[l, Sequence @@ Table[x, {n}], r];
UU[l___, 1, r___] := UU[l, r];
UU[] = U[]; UU[l_, r___] := U[l] ** UU[r];
U_i_[ℰ_] := ℰ /. {t → t_i, u_U ⧴ UU @@ Replace[u, x_ ⧴ x_i, 1]};
```

```
B[x_, x_] = 0;
B[U[(x_)_i_], U[(y_)_i_]] := U_i[B[U@x, U@y]];
B[U[(x_)_i_], U[(y_)_j_]] /; i =!= j := 0;
B[x_, y_] := x ** y - y ** x;
```

```
x_ ≤ y_ := OrderedQ[{x, y}]; x_ < y_ := ! OrderedQ[{y, x}];
Simp[ℰ_] := Collect[ℰ, _U, Together];
```

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
0 ** _ = _ ** 0 = 0;
x_ ** U[] := x; U[] ** x_ := x;
(a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, Simp[a b (x ** y)]];
(a_ * x_U) ** y_ := Simp[a (x ** y)]; x_ ** (a_ * y_U) := Simp[a (x ** y)];
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
U[x_] ** U[y_] := (*U[x]**U[y] =*) If[x < y, U[x, y], U[y, x] + B[U@x, U@y]];
U[x_] ** U[y1_, yy__] := (*U[x] ** U[y1,yy] =*)
  If[x ≤ y1, U[x, y1, yy], (U@x ** U@y1) ** U@yy];
U[xx__, xn_] ** U[yy__] := (*U[xx,xn]**U[yy] =*) U@xx ** (U@xn ** U@yy);
```

```
σ[i_, j_][ℰ_] := ℰ /. {b_i → b_j, t_i → t_j, a_i → a_j, c_i → c_j, u_i → u_j, w_i → w_j};
```

```
mul[i_, j_][ℰ_] :=
  Simp[ℰ /. x_U ⧴ DeleteCases[x, _j] ** U @@ Cases[x, y_j ⧴ y_i] /. {b_j → b_i, t_j → t_i}];
mul[i_, j_, k_][ℰ_] := ℰ // mul[i, j] // σ[i, k];
```

```
Δ[i_, j_, k_][ℰ_] := Simp[ℰ /. {
    z_. U[] ⧴ (z /. {b_i → b_j + b_k, t_i → t_j t_k}) U[],
    z_. x_U ⧴ (z /. {b_i → b_j + b_k, t_i → t_j t_k}) NonCommutativeMultiply @@ (x /. {
         c_i → U@c_j + U@c_k,
         u_i → t_k U@u_j + ϵ t_k UU[c_k, u_j] + U@u_k,
         w_i → U@w_j + ϵ UU[c_k, w_j] + U@w_k,
         a_i → t_k U[a_j] + U[a_k] - ϵ U[a_k, c_j],
         y_l_ ⧴ U@y_l
        })
   }]
```

```
S[i_][ℰ_] := Simp[ℰ /. {z_. x_U ⧴ (z /. {b_i → -b_i, t_i → t_i^-1}) S[i][x]}];
S[i_][U[]] = U[];
S[i_][U[y_j_, more___]] /; i ≠ j := U[y_j] ** S[i][U[more]];
(*Careful! if mathematica cannot decide i==j or not then we get an error*)
S[i_][U[c_i_, more___]] := S[i][U[more]] ** (-U@c_i);
S[i_][U[u_i_, more___]] := S[i][U[more]] ** (-t_i^-1 U@u_i + Expand[ϵ t_i^-1 UU[u_i, c_i]]);
S[i_][U[w_i_, more___]] := S[i][U[more]] ** (-U@w_i + Expand[ϵ UU[w_i, c_i]]);
S[i_][U[a_i_, more___]] := S[i][U[more]] ** (Expand[-t_i^-1 ((1 + ϵ) U[a_i] + ϵ U[a_i, c_i])]);
```

```
Sinv[i_][ℰ_] := Simp[ℰ /. {z_. x_U ⧴ (z /. {b_i → -b_i, t_i → t_i^-1}) Sinv[i][x]}];
Sinv[i_][U[]] = U[];
Sinv[i_][U[y_j_, more___]] /; i ≠ j := U[y_j] ** Sinv[i][U[more]];
(*Careful! if mathematica cannot decide i==j or not then we get an error*)
Sinv[i_][U[c_i_, more___]] := Sinv[i][U[more]] ** (-U@c_i);
Sinv[i_][U[u_i_, more___]] :=
  Sinv[i][U[more]] ** (-t_i^-1 U@u_i + Expand[ϵ t_i^-1 UU[c_i, u_i]]);
Sinv[i_][U[w_i_, more___]] := Sinv[i][U[more]] ** (-U@w_i + Expand[ϵ UU[c_i, w_i]]);
Sinv[i_][U[a_i_, more___]] := Sinv[i][U[more]] ** (Expand[-t_i^-1 (U[a_i] + ϵ U[a_i, c_i])]);
```

```
CoUnit[i_][ℰ_] :=
  Simp[ℰ /. {z_. x_U ⧴ (z /. {t → 1, b → 0, b_i → 0, t_i → 1}) CoUnit[i][x]}] ;
CoUnit[i_][U[]] = U[];
CoUnit[i_][U[y_j_, more___]] /; i ≠ j := U[y_j] ** CoUnit[i][U[more]];
CoUnit[i_][U[y_i_, more___]] /; y ≠ a := 0;
CoUnit[i_][U[a_i_, more___]] := CoUnit[i][U[more]];
```

```
LBasis[n_Integer] := LBasis[Range[n]];
LBasis[S_] := DeleteCases[0]@
  Module[{i, j, k, l}, SortBy[(# /. {ϵ → 2, c_ → 2, u_ → 2, w_ → 2, U → Times}) &][
    Union@Flatten[{{U[], ϵ U[]},
        Table[{U@c_i, U@u_i, U@w_i, ϵ U@c_i, ϵ U@u_i, ϵ U@w_i}, {i, S}],
        Table[{U[u_i, w_j], ϵ U[u_i, w_j],
          ϵ U @@ Sort@{c_i, c_j}, ϵ U[c_i, u_j], ϵ U[c_i, w_j]}, {i, S}, {j, S}],
        Table[{ϵ U[c_i, u_j, w_k], ϵ U @@ Sort@{u_i, u_j, w_k}, ϵ U @@ Sort@{u_i, w_j, w_k}},
         {i, S}, {j, S}, {k, S}],
        Table[ϵ U @@ Sort@{u_i, u_j, w_k, w_l}, {i, S}, {j, S}, {k, S}, {l, S}] }]
   ]]
```

```
BLBasis[n_Integer] := BLBasis[Range[n]];
BLBasis[S_] := DeleteCases[0]@
  Module[{i, j, k, l}, SortBy[(# /. {ϵ → 2, c_ → 2, u_ → 2, w_ → 2, U → Times}) &][
    Union@Flatten[{{U[], ϵ U[]},
        Table[{U@c_i, ϵ U@c_i}, {i, S}],
        Table[{U[u_i, w_j], ϵ U[u_i, w_j], ϵ U @@ Sort@{c_i, c_j}}, {i, S}, {j, S}],
        Table[{ϵ U[c_i, u_j, w_k]}, {i, S}, {j, S}, {k, S}],
        Table[ϵ U @@ Sort@{u_i, u_j, w_k, w_l}, {i, S}, {j, S}, {k, S}, {l, S}] }]
   ]]
```

```
UExp[ℰ_, n_] := Module[{t = U[], k}, U[] + Sum[(t = t ** ℰ)/k!, {k, n}]] // Simp
```

```
ToDegree[n_][ℰ_] :=
 (Simp[ℰ] /. {ϵ → ℏ ϵ, η → ℏ η, b_i_ :→ ℏ b_i, t_i_ :→ e^(ℏ b_i), b → ℏ b, t :→ e^(ℏ b), x_U :→
      ℏ^(Count[x,u|u_]+Count[x,a|a_]) x} /. a_. x_U :→ Normal[Series[a, {ℏ, 0, n}]] * x) /. ℏ → 1
```

Hopf algebra axioms

```
(*Sinv is the inverse of S *)
Sinv[1][S[1][U@a_1]] // Simp
Sinv[1][S[1][U@c_1]] // Simp
Sinv[1][S[1][U@w_1]] // Simp
Sinv[1][S[1][U@u_1]] // Simp
```

```
U[a_1]

U[c_1]

U[w_1]

U[u_1]
```

```
(*Coassociativity only works properly with tᵢ instead of t!*)
(U@w1 // Δ[1, x, yy] // Δ[yy, y, z]) -
 (U@w1 // Δ[1, xx, z] // Δ[xx, x, y])
(U@c1 // Δ[1, x, yy] // Δ[yy, y, z]) -
 (U@c1 // Δ[1, xx, z] // Δ[xx, x, y])
(U@a1 // Δ[1, x, yy] // Δ[yy, y, z]) -
  (U@a1 // Δ[1, xx, z] // Δ[xx, x, y]) // Simp
(U@u1 // Δ[1, x, yy] // Δ[yy, y, z] // Simp) -
  (U@u1 // Δ[1, xx, z] // Δ[xx, x, y] // Simp) // Simp
```

0

0

0

0

```
(*Convolution inverse*)
mul[2, 3, 1][S[2][Δ[1, 2, 3][U[w1]]]]
mul[2, 3, 1][S[2][Δ[1, 2, 3][U[c1]]]]
mul[2, 3, 1][S[2][Δ[1, 2, 3][U[u1]]]]
mul[2, 3, 1][S[2][Δ[1, 2, 3][U[a1]]]]

mul[2, 3, 1][S[3][Δ[1, 2, 3][U[w1]]]]
mul[2, 3, 1][S[3][Δ[1, 2, 3][U[c1]]]]
mul[2, 3, 1][S[3][Δ[1, 2, 3][U[u1]]]]
mul[2, 3, 1][S[3][Δ[1, 2, 3][U[a1]]]]
```

0

0

0

0

0

0

0

0

Testing Yang-Baxter

```
R[i_, j_, d_] :=
 Sum[1/(m! n!) UU[a_i^n] ** (U[] + ϵ n U[c_i]) ** (b_i^m U[] + ϵ m b_i^(m-1) U[c_i]) ** UU[c_j^m, w_j^n]
   (1 + ϵ 1/4 (n - 1) n), {m, 0, d}, {n, 0, d - m}] // ToDegree[d]
```

```
RU[i_, j_, d_] := Sum[1/(m! n!) UU[u_i^n] ** (b_i^m U[] + ϵ m b_i^(m-1) U[c_i]) ** UU[c_j^m, w_j^n] (1 - ϵ/4 (n - 1) n),
   {m, 0, d}, {n, 0, d - m}] // ToDegree[d]
```

```
RU3[d_] := (ToDegree[d][RU[1, 2, d] ** RU[1, 3, d]] ** RU[2, 3, d]) -
  (ToDegree[d][RU[2, 3, d] ** RU[1, 3, d]] ** RU[1, 2, d])
```

```
Timing[RU3[2] // ToDegree[2]]
```

{0.335015, 0}

```
Timing[RU3[3] // ToDegree[3]]
```

{5.76008, 0}

```
Timing[RU3[4] // ToDegree[4]]
```

{125.708, 0}

```
R3[d_] := (ToDegree[d][R[1, 2, d] ** R[1, 3, d]] ** R[2, 3, d]) -
  (ToDegree[d][R[2, 3, d] ** R[1, 3, d]] ** R[1, 2, d])
```

```
R3[2] // ToDegree[2]
```

0

```
Timing[R3[3] // ToDegree[3]]
```

{4.55927, 0}

```
Timing[R3[4] // ToDegree[4]]
```

{49.7652, 0}

```
(*Verifying the inverse is given by (S tensor id)(R)*)
(S[1][R[1, 2, 4]] ** R[1, 2, 4] - U[]) // ToDegree[4]
(*Verifying the inverse is given by (id tensor S^{-1})(R)   *)
```

0

```
(Sinv[2][R[1, 2, 4]] ** R[1, 2, 4] - U[]) // ToDegree[4]
```

0

```
Rinv[i_, j_, d_] := S[i][R[i, j, d]] // Expand
```

```
(*Alternative formulas for Rinv, Rinv4 is the definitive one*)
```

Quasi triangularity axioms

```
(*Check the three quasi-triangularity axioms*)
(Δ[i, k, l][R[i, j, 5]] - R[k, j, 5] ** R[l, j, 5]) // ToDegree[5]
```

0

```
(Δ[j, k, l][R[i, j, 5]] - R[i, l, 5] ** R[i, k, 5]) // ToDegree[5]
```

0

```
CheckRDR[x_, d_] :=
 (R[2, 3, d] ** Δ[1, 2, 3][x] ** Rinv[2, 3, d] - σ[2, 3][Δ[1, 2, 3][x]]) // ToDegree[d]
```

```
CheckRDR[U@c_i, 4]
CheckRDR[U@w_i, 4]
CheckRDR[U@a_i, 4]
```

0

0

0

Drinfeld element

```
(*Drinfeld element*)
Dr[d_] := R[1, 2, d] // S[2] // mul[2, 1, 1]
```

```
(*Check that S(Dr) and Dr commute*)
S[1][Dr[4]] ** Dr[4] - S[1][Dr[4]] ** Dr[4]
```

0

```
S[1][Dr[2]] // ToDegree[2]
```

$$U[\,] - b_1 U[c_1] + (-1+\epsilon+b_1)\, U[a_1, w_1] + \left(-\epsilon+\frac{b_1^2}{2}\right) U[c_1, c_1] + (\epsilon+b_1)\, U[a_1, c_1, w_1] + \epsilon\, b_1 U[c_1, c_1, c_1] + \frac{1}{2} U[a_1, a_1, w_1, w_1] + \epsilon\, U[a_1, c_1, c_1, w_1]$$

```
(*S(Dr) = t^-1 e^-2ϵc Dr *)
S[1][Dr[5]] - (U[] - 2 ϵ U[c_1]) ** (Dr[5] t_1^-1 // ToDegree[5]) // ToDegree[5]
```

0

```
(U[] - ϵ U[c_1]) ** Dr[4] - Dr[4] ** (U[] - ϵ U[c_1]) // ToDegree[4]
```

0

Therefore the Ribbon element v is implied by $v^2 = \mathrm{S(Dr)Dr} = t^{-1}\, e^{-2\,\epsilon\, c}\, \mathrm{Dr}^2$ so choose $v = t^{-1/2}\, e^{-\epsilon\, c}\, \mathrm{Dr}$, note Dr commutes with $e^{-\epsilon c}$.

According to Ohtsuki p.72 read upside down, we should set the left-moving cup and cap to 1 and the right-moving cap nr should be $v\,\mathrm{Dr}^{-1}=\mathrm{t}^{-1/2}\,\mathrm{e}^{-\epsilon\,\mathrm{c}}$ and
the right-moving cup ur should be $\mathrm{Dr}\,v^{-1}=t^{1/2}\,e^{\epsilon c}$.

Logos

```
B[U@w, U@a]
```

$(1-t)\,U[\,]-(1+t)\,\epsilon\,U[c]$

```
(*η /: η^n_/;n>1 :=0; *)
η = (t + 1)/(t - 1) ϵ;
q = 1 + η;
qI[k_] := k (1 + η (k - 1) / 2) // Expand
qFac[n_] := n! (1 + η (n - 1) n / 4) // Expand
InvqFac[n_] := (1 - η (n - 1) n / 4) / n! // Expand
qBin[n_, k_] := Binomial[n, k] (1 + η k (n - k) / 2) // Expand
```

```
(*Checking the commutation relation for powers of a,w*)
WmAn[m_, n_] := Sum[(1 - t)^j qBin[m, j] qBin[n, j]
   qFac[j] UU[a^(n-j)] ** (U[] + j η U[c]) ** UU[w^(m-j)], {j, 0, Min[m, n]}]
TestWmAn[m_, n_] := -UU[w^m, a^n] + WmAn[m, n]
TestWmAn[5, 3] // Together
```

0

```
(*Guess qLogos first at q=1*)
ToDegh[F_, x_] := Series[F /. {α → α h, δ → δ h, β → β h}, {h, 0, x}]
d = 8;
LHS = Sum[α^m δ^k β^n InvqFac[m] InvqFac[k] InvqFac[n] WmAn[m + k, n + k],
    {m, 0, d}, {k, 0, d - m}, {n, 0, d - m - k}] // Simp;
(*powers of nu as power series*)
nuA[z_] := Sum[Binomial[z - 1 + x, x] (1 - t)^x δ^x, {x, 0, d}];
RHS =
  Sum[(1 - t)^j nuA[m + k + n + j + 1] α^(m+j) δ^k β^(n+j) InvqFac[m] InvqFac[k] InvqFac[n] InvqFac[j]
    UU[a^(n+k), w^(m+k)], {m, 0, d}, {k, 0, d - m}, {n, 0, d - m - k}, {j, 0, d - m - n - k}] // Simp;
ToDegh[(LHS - RHS) /. {ϵ → 0} // Simp, 8]
```

$O[h]^9$

```
(*Now set up the LHS and RHS to find the Logos relating them.*)
d = 11;
```

$$\text{LHS} = \text{Sum}\left[\alpha^m \delta^k \beta^n \frac{\text{WmAn}[m+k, n+k]}{m!\,k!\,n!}, \{m, 0, d\}, \{k, 0, d-m\}, \{n, 0, d-m-k\}\right] \text{ // Simp;}$$

$$\text{RHS} = \text{Sum}\left[(1-t)^j \mu^{-(m+k+n+j+1)} \frac{\alpha^{m+j} \delta^k \beta^{n+j}}{m!\,k!\,n!\,j!} \text{UU}\left[a^{n+k}, w^{m+k}\right], \{m, 0, d\}, \{k, 0, d-m\}, \{n, 0, d-m-k\}, \{j, 0, d-m-n-k\}\right];$$

$$\text{CRHS} = \text{Sum}\left[(1-t)^j \mu^{-(m+k+n+j+1)} \frac{\alpha^{m+j} \delta^k \beta^{n+j}}{m!\,n!\,k!\,j!} \text{UU}\left[a^{n+k}, c, w^{m+k}\right], \{m, 0, d\}, \{k, 0, d-m\}, \{n, 0, d-m-k\}, \{j, 0, d-m-n-k\}\right] \text{ // Simp;}$$

$$\mu = 1 - (1-t)\,\delta;$$

$$\eta$$

$$\frac{(1+t)\,\epsilon}{-1+t}$$

```
(*Here we guess and verify the Logos coefficient by coefficient*)
```

$$\text{ToDegh}\Big[\text{Coefficient}\Big[\Big(-\text{LHS} + \text{RHS} + \mu^{-4}(1-t)\,\eta\,\Big((1-t)\left(\frac{1}{2}\delta^2\mu^2 + \alpha\beta\delta\mu + \frac{\alpha^2\beta^2}{4}\right)\text{RHS} + \mu^2(\delta\mu+\alpha\beta)\,\text{CRHS} + \beta\left(\delta\mu + \frac{\alpha\beta}{2}\right)\text{U[a] ** RHS} + \alpha\left(\delta\mu+\frac{\alpha\beta}{2}\right)\text{RHS ** U[w]} + \beta\delta\mu^2\,\text{U[a] ** CRHS} + \alpha\delta\mu^2\,\text{CRHS ** U[w]} + \delta(1+\mu)(\mu\delta+\alpha\beta)\,\text{U[a] ** RHS ** U[w]} + \delta^2\mu^2\,\text{U[a] ** CRHS ** U[w]} + \frac{\beta^2\delta}{4}(1+\mu)\,\text{U[a, a] ** RHS} + \frac{\alpha^2\delta}{4}(1+\mu)\,\text{RHS ** U[w, w]} + \frac{\beta\,\delta^2}{2}(1+2\mu)\,\text{U[a, a] ** RHS ** U[w]} + \frac{\alpha\,\delta^2}{2}(1+2\mu)\,\text{U[a] ** RHS ** U[w, w]} + \frac{\delta^3}{4}(1+3\mu)\,\text{U[a, a] ** RHS ** U[w, w]}\Big)\Big), \text{U[]}\Big], 10\Big]$$

$$O[h]^{11}$$

```
(*Final check, put back ϵ.*)
ToDegh[(-LHS + RHS +
    -μ^-4 (1 + t) ϵ (
      (1 - t) (1/2 δ^2 μ^2 + α β δ μ + α^2 β^2/4) RHS +
      μ^2 (δ μ + α β) CRHS +
      β (δ μ + α β/2) U[a] ** RHS +
      α (δ μ + α β/2) RHS ** U[w] +
      β δ μ^2 U[a] ** CRHS +
      α δ μ^2 CRHS ** U[w] +
      δ (1 + μ) (μ δ + α β ) U[a] ** RHS ** U[w] +
      δ^2 μ^2 U[a] ** CRHS ** U[w] +
      β^2 δ/4 (1 + μ ) U[a, a] ** RHS +
      α^2 δ/4 (1 + μ ) RHS ** U[w, w] +
      β δ^2/2 (1 + 2 μ) U[a, a] ** RHS ** U[w] +
      α δ^2/2 (1 + 2 μ) U[a] ** RHS ** U[w, w] +
      δ^3/4 (1 + 3 μ ) U[a, a] ** RHS ** U[w, w])
  ), 10] // Simp
0
```